HAI 1.2

    HOW IZ I SUMDIVISORS YR N
        I HAS A D ITZ 2
        I HAS A S ITZ 0
        IM IN YR LOOP UPPIN YR NOTIN WILE DIFFRINT SMALLR OF PRODUKT OF D D N N
            BOTH SAEM MOD OF N D 0, O RLY?
                YA RLY, S R SUM OF S SUM OF D QUOSHUNT OF N D
            OIC
            D R SUM OF D 1
        IM OUTTA YR LOOP
        BOTH SAEM PRODUKT OF D D N, O RLY?, YA RLY, S R SUM OF S D, OIC
        FOUND YR SUM OF S 1
    IF U SAY SO

    I HAS A FRIENDSHIP ITZ 0
    I HAS A M
    IM IN YR LOOP UPPIN YR N WILE DIFFRINT N 10001
        M R I IZ SUMDIVISORS YR N MKAY
        ALL OF...
            BOTH SAEM I IZ SUMDIVISORS YR M MKAY N...
            DIFFRINT SMALLR OF M 10000 10000...
            DIFFRINT M N...
        MKAY
        O RLY?
            YA RLY, FRIENDSHIP R SUM OF FRIENDSHIP N
        OIC
    IM OUTTA YR LOOP
    VISIBLE FRIENDSHIP

KTHXBYE
